Maple-assisted proof of formula for A185818

Robert Israel

10 July 2018

There are $5^4 = 625$ configurations for a 2×2 sub-array.. We encode these configurations as lists in the order

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}.$$

Consider the 625×625 transition matrix T with entries $T_{ij} = 1$ if the first two rows of a 3×2 sub-array could be in configuration i while the last two rows are in configuration j (so in particular the second row of configuration i is the first row of configuration j), and 0 otherwise. The following Maple code computes it.

```
> Configs:=  [seq(convert(5^4+i,base,5)[1..4],i=0..5^4-1)]:
> Compatible:= proc(i,j) local k;
   if Configs[i][3..4] <> Configs[j][1..2] then return 0 fi;
   k:= numboccur(Configs[i][3],[Configs[i][1],Configs[i][4],Configs
  [j][3]]);
   if k = 0 or k = 3 then return 0 fi;
   k:= numboccur(Configs[i][4],[Configs[i][2],Configs[i][3],Configs
  [j][4]]);
   if k = 0 or k = 3 then 0 else 1 fi;
  end proc:
  T:= Matrix(625,625,Compatible):
```

Thus for $n \geq 2$ $a(n) = \dfrac{u^T\, T^{n-2} v}{5}$ where u and v are column vectors such that $u_i = 1$ if the top two rows of the array could be in configuration i and 0 otherwise, i.e. each of the top row entries of configuration i is equal to one or two of its neighbours in that configuration, and similarly for v with the bottom two rows of the array. The following code computes them.

```
> u:= Vector(625, proc(i) if Configs[i][1]=Configs[i][2] or
  (Configs[i][1]=Configs[i][3] and
    Configs[i][2]=Configs[i][4]) then 1 else 0 fi end proc):
  v:= Vector(625, proc(i) if Configs[i][3]=Configs[i][4] or
  (Configs[i][1]=Configs[i][3] and
    Configs[i][2]=Configs[i][4]) then 1 else 0 fi end proc):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 23 do TV[n]:= T . TV[n-1] od:
> A:= [1,seq(u^%T . TV[n]/5,n=0..23)];
```

A := [1, 9, 76, 656, 5680, 49248, 426928, 3701360, 32089696, 278208816, 2411993584, 20911320416, 181295389360, 1571781109104, 13626909445216, 118141552910384, 1024254735084784, 8880006538838880, 76987211704914352, 667457928119357552, 5786676461500240480, 50168891640003776304, 434950477209487093744, 3770900879817532340576, 32692672362697386211504] **(1)**

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{20} - 2\,t^{19} - 36\,t^{18} - 107\,t^{17} - 481\,t^{16} - 1514\,t^{15} - 3338\,t^{14} - 6913\,t^{13} - 11193\,t^{12} - 10312\,t^{11} + 16852\,t^{10} + 84423\,t^{9} + 155068\,t^{8} + 170768\,t^{7} + 88064\,t^{6} - 26624\,t^{5} - 114688\,t^{4} - 159744\,t^{3} - 114688\,t^{2} - 65536\,t \quad (2)$$

```
> degree(P(t));
```

$$20 \quad (3)$$

This turns out to have degree 20, but with the t^0 coefficient 0. Thus we will have

$0 = u^T P(T)\, T^n\, v = \sum_{i=1}^{20} p_i\, a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 19, which would hold true for any u and v, after a delay of 1. It seems that with our particular u and v we have a recurrence of order only 5, corresponding to a factor of P.

```
> n:= 'n':
  empirical:= a(n)=7*a(n-1) + 15*a(n-2) - 32*a(n-4) - 64*a(n-5):
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(5-i),i=0..5),t);
```

$$Q := t \mapsto t^5 - 7\,t^4 - 15\,t^3 + 32\,t + 64 \quad (4)$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 15, again with the lowest two coefficients 0.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(t^{14} + 5\,t^{13} + 14\,t^{12} + 66\,t^{11} + 159\,t^{10} + 365\,t^{9} + 834\,t^{8} + 1392\,t^{7} + 1749\,t^{6} + 955\,t^{5} - 276\,t^{4} - 1104\,t^{3} - 1856\,t^{2} - 1280\,t - 1024\right) t \quad (5)$$

```
> degree(R(t));
```

$$15 \quad (6)$$

Now we want to show that $c(n) = u^T Q(T)\, T^n\, v = 0$ for all $n \geq 0$. This will certainly satisfy the recurrence

$$\sum_{i=1}^{15} r_i\, c(i+n) = \sum_{i=1}^{15} r_i u^T Q(T)\, T^{n+i} v = u^T Q(T)\, R(T)\, T^n\, v = u^T P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(0) = ... = c(14) = 0$.

```
> UT[0]:= u^%T:
  for n from 1 to 14 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=0..5):
> seq(w . TV[n],n=0..14);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \quad (7)$$

This completes the proof.